
;AUTHOR:   MUHAMMAD WAQAR ASHRAF 
;DATE:   12/06/18 
;DRUG/S:   S-KETAMINE + NORKETAMINE + TICLOPIDINE 
;MODEL:   KETAMINE + NORKETAMINE + TICLOPIDINE + DRUG-DRUG-INTERACTION MODEL 
 
;------------------------------------------------------------------------  
$PROBLEM    KETAMINE/NORKETAMINE/TICLOPIDINE/INTERACTION MODEL 
 
;------------------------------------------------------------------------  
$INPUT      ID PHASE=DROP TICLO FLAG=DROP TIME DV AMT ADDL II CMT RATE MDV EVID 
WTKG SEXF=DROP 
 
;------------------------------------------------------------------------ 
$DATA       data.csv IGNORE=@ 
 
;------------------------------------------------------------------------ 
$SUBROUTINE ADVAN13 TRANS1 TOL=6 
 
;------------------------------------------------------------------------ 
; UNITS 
;  Time - hours 
;  Dose - mg 
;  Cp   - ng/ml = mcg/L 
;  Clearances - L/hour 
;  Volumes - L 
 
;------------------------------------------------------------------------ 
$MODEL 
COMP=(KDEPOT)                         ; 1 Ketamine depot 
COMP=(CENTRAL)                        ; 2 Ketamine central 
COMP=(PERIPH1)                        ; 3 Ketamine peripheral shallow 
COMP=(PERIPH2)                        ; 4 Ketamine peripheral deep 
COMP=(NKCENT)                         ; 5 Norketamine central 
COMP=(PERIM)                          ; 6 Norketamine peripheral 
COMP=(TDEPOT)                         ; 7 Ticlopidine depot 
COMP=(TGUTWALL)                       ; 8 Ticlopidine gutwall 
COMP=(TPV)                            ; 9 Ticlopidine portal vein 
COMP=(TLIVER)                         ; 10 Ticlopidine liver 
COMP=(TCENT)                          ; 11 Ticlopidine central 
COMP=(TPERIPH)                        ; 12 TIclopidine peripheral 
COMP=(TT1)                            ; 13 Ticlopidine transit 1 
COMP=(TT2)                            ; 14 Ticlopidine transit 2 
COMP=(TT3)                            ; 15 Ticlopidine transit 3 
COMP=(TT4)                            ; 16 Ticlopidine transit 4 
 
;------------------------------------------------------------------------ 
$PK 
; Ketamine model 
 
KV1 = THETA(1)                        ; Ketamine: volume of central comp. 
KQ = THETA(2)                         ; Ketamine: central to peripheral clearance 
KV2 = THETA(3)*EXP(ETA(1))            ; Ketamine: volume of periph comp. 
KKA = THETA(4)*EXP(ETA(2))            ; Ketamine: absorption rate constant 
KCLINTH1 = THETA(5)*EXP(ETA(3))       ; Ketamine: intrinsic hepatic clearance 
KV3 = THETA(6)                        ; Ketamine: volume of peripheral 2 
KQ2 = THETA(7)                        ; Ketamine: clearance to peripheral 2 
KCLINTG1 = THETA(8)*EXP(ETA(4))       ; Ketamine: intrinsic gutwall clearance 
KCLP = THETA(9)                       ; Ketamine: permeability clearance 
 
;------------------------------------------------------------------------ 
; Norketamine model 
 
NKV5 = THETA(10)                      ; Norketamine: volume of central 
NKQ3 = THETA(11)                      ; Norketamine: central to peripheral cl 



NKV6 = THETA(12)                      ; Norketamine: volume of peripheral 
NKCLINTH = THETA(13)*EXP(ETA(5))      ; Norketamine: hepatic cl of metabolite 
NKCLINTG = THETA(14)                  ; Norketamine: gutwall cl of metabolite 
 
;----------------------------------------------------------------------------------  
; Ticlopidine model 
 
TKA = THETA(15)                       ; Ticlopidine: fast absorption rate constant 
TCLINTH = THETA(16)*EXP(ETA(6))       ; Ticlopidine: intrinsic hepatic clearance 
TV1  = THETA(17)                      ; Ticlopidine: volume of central 
TQ = THETA(18)                        ; Ticlopidine: central to peripheral cl 
TV2 = THETA(19)                       ; Ticlopidine: volume of peripheral 
TCLINTG = THETA(20)                   ; Ticlopidine: gutwall clearance 
 
;----------------------------------------------------------------------------------  
; Interaction model 
 
TKI = THETA(21)                       ; KI of ticlopidine for CYP2B6 
TKINACT = THETA(22)                   ; Inact rate of enzyme CYP2B6 for ticlopidine 
KFM2B6 = THETA(23)                    ; Fraction metabolized by CYP2B6 
LAMBDA = THETA(24)                    ; Rate of CYP2B6 degradation at ticlo = 1 
TKIN = THETA(25)                      ; IC50 of ticlo on CYP2B6  
 
;----------------------------------------------------------------------------------  
; Drug constants 
 
KFU = 0.70                            ; Ketamine: fraction unbound in blood 
NKFUB = 0.50                          ; Norketamine: fub of norketamine in plasma 
TFU = 0.02                            ; Ticlopidine: fraction unbound in blood 
KRATIO = 0.50                         ; blood plasma ratio (Launiainen 2014) 
KFUB = KFU/KRATIO                     ; fraction unbound in the blood 
TRATIO = 1                            ; Ticlopidine ratio assumption = 1 
TFUB = TFU/TRATIO                     ; Ticlopidine fraction unbound in blood 
KFUG = 1                              ; fixed to 1, acc to Yang et al. 2007 
TFUG = 1                              ; fixed to 1 for Ticlopidine 
NKFUG = 1                             ; Norketamine: fub of norketamine in gutwall 
 
;----------------------------------------------------------------------------------  
; Ticlopidine absorption 
F7 = 1 
ALAG7 = 0 
TKTR = TKA 
 
; Ketamine absorption 
F1 = 1 
ALAG1 = 0 
D2 = 0.03333                         ; duration of ketamine infusion = 2 minutes 
 
;----------------------------------------------------------------------------------  
; Physiological parameters 
 
QH = 3.75*WTKG**0.75                 ; in L/HR (according to Brown et al. 1997) 
QPV = 0.75*QH                        ; portal vein blood flow. Williams et al. 1989 
QHA = 0.25*QH                        ; hepatic art blood flow. Williams et al. 1989 
QIN = 0.4*QH                         ; intestial blood flow, Williams et al. 1989 
QMU = 0.8*QIN                        ; mucosa blood flow, Yang et al. 2007 
QVI = 0.6*QMU                        ; villous blood flow, Yang et al. 2007 
QGUT = (QVI*KCLP)/(QVI+KCLP)         ; QGUT model for Ketamine 
VGW = 1                              ; volume of Gut wall fixed to 1 L 
VPV = 1                              ; volume of portal vein fixed to 1 L 
VH = 1                               ; volume of liver fixed to 1 L 
 
;----------------------------------------------------------------------------------  
; Gutwall well stirred clearance model Ketamine 



 
CLINTG = KCLINTG1 
FG = QGUT/(QGUT+(CLINTG*KFUG)) 
EG = 1 - FG 
CLG = QGUT*KFUG*CLINTG/(QGUT*KFUG+CLINTG) 
 
;----------------------------------------------------------------------------------  
; Gutwall well stirred clearance model Norketamine 
 
CLINTGM = NKCLINTG 
FGM = QVI/(QVI+(CLINTGM*NKFUG)) 
EGM = 1 - FGM 
CLGM = QVI*NKFUG*CLINTGM/(QVI*NKFUG+CLINTGM) 
 
;----------------------------------------------------------------------------------  
; Hepatic well stirred clearance model Norketamine 
 
FHM = QH/(QH+(NKCLINTH*NKFUB)) 
EHM = 1 - FHM 
CLHM = QH*NKFUB*NKCLINTH/(QH+NKFUB*NKCLINTH) 
 
;----------------------------------------------------------------------------------  
; Hepatic well stirred clearance ticlopidine 
 
TFH = QH/(QH+(TCLINTH*TFUB)) 
TEH = 1 - TFH 
TCLH = QH*TCLINTH*TFUB/(QH+TCLINTH*TFUB) 
 
;----------------------------------------------------------------------------------  
; Gutwall well stirred clearance model ticlopidine 
 
TEG = (TCLINTG*TFUG)/(QVI+(TCLINTG*TFUG)) 
TFG = 1 - TEG 
TCLG = QVI*TFUG*TCLINTG/(QVI+TFUG*TCLINTG) 
 
;----------------------------------------------------------------------------------  
; Three compartmental model Ketamine 
 
K23 = KQ/KV1 
K32 = KQ/KV2 
K24 = KQ2/KV1 
K42 = KQ2/KV3 
 
;----------------------------------------------------------------------------------  
; Two compartmental model Norketamine 
 
K56 = NKQ3/NKV5 
K65 = NKQ3/NKV6 
 
;----------------------------------------------------------------------------------  
; Two compartmental model Ticlopidine 
 
K1112 = TQ/TV1 
K1211 = TQ/TV2 
 
;----------------------------------------------------------------------------------  
S2 = KV1/1000 
S5 = NKV5/1000 
S11 = TV1/1000 
 
;----------------------------------------------------------------------------------  
$DES 
 
TCE = TFUB*(A(9)+((F7*TKA*AMT)/QH))                ; Inh conc in the portal vein 



TAH = 1/(1 + (TCE/TKIN))                           ; reversible inhibition 
TBH = LAMBDA/(LAMBDA + (TKINACT*TCE/(TKI+TCE)))    ; time depend. inhibition 
ZETA = (TAH * TBH * KFM2B6) + (1 - KFM2B6)         ; inh param comp and mech models 
 
; Hepatic Clearance Ketamine 
; Inhibition model 
 
INHBH = 1 
CLINTH = KCLINTH1 
IF (TICLO.EQ.1) THEN 
INHBH = ZETA                      ; inhib param CLINTH of Ket at ticlo = 1 
CLINTH = KCLINTH1 * INHBH         ; reduction in clearance of ketamine at ticlo = 1 
ENDIF 
 
; Well stirred clearance model 
 
EH = (CLINTH*KFUB)/(QH+(CLINTH*KFUB)) 
FH = 1 - EH 
CLH = QH*KFUB*CLINTH/(QH+KFUB*CLINTH) 
 
DADT(1) = -(KKA*A(1)) 
 
AGUTW = KKA*A(1)/(QVI/VGW) 
APV = ((QVI/VGW)*FG*AGUTW + (QPV/KV1)*A(2))/(QPV/VPV) 
AH = ((QHA/KV1)*A(2) + (QPV/VPV)*APV)/(QH/VH) 
 
DADT(2) = FH*(QH/VH)*AH - (QHA/KV1)*A(2) - (QPV/KV1)*A(2) - K23*A(2) + K32*A(3) - 
K24*A(2) + K42*A(4) 
DADT(3) = K23*A(2) - K32*A(3) 
DADT(4) = K24*A(2) - K42*A(4) 
 
AGUTWM = (1-FG)*AGUTW 
APVM = ((QVI/VGW)*AGUTWM*FGM + (QPV/NKV5)*A(5))/(QPV/VPV) 
AHM = ((QHA/NKV5)*A(5) + (QPV/VPV)*APVM + (QH/VH)*EH*AH)/(QH/VH) 
 
DADT(5) = FHM*(QH/VH)*AHM - (QPV/NKV5)*A(5) - (QHA/NKV5)*A(5) - K56*A(5) + K65*A(6) 
DADT(6) = K56*A(5) - K65*A(6) 
 
DADT(7) = -(TKA*A(7)) 
 
DADT(8) =  TKTR*A(16) - (QVI/VGW)*TEG*A(8) - (QVI/VGW)*TFG*A(8) 
DADT(9) =  (QVI/VGW)*TFG*A(8) + (QPV/TV1)*A(11) - (QPV/VPV)*A(9) 
DADT(10) = (QPV/VPV)*A(9) + (QHA/TV1)*A(11) - TEH*(QH/VH)*A(10) - TFH*(QH/VH)*A(10) 
DADT(11) = TFH*(QH/VH)*A(10) - (QPV/TV1)*A(11) - (QHA/TV1)*A(11) - K1112*A(11) + 
K1211*A(12) 
DADT(12) = K1112*A(11) - K1211*A(12) 
 
DADT(13) = TKA*A(7) - TKTR*A(13) 
DADT(14) =            TKTR*A(13) - TKTR*A(14) 
DADT(15) =                         TKTR*A(14) - TKTR*A(15) 
DADT(16) =                                      TKTR*A(15) - TKTR*A(16) 
 
;----------------------------------------------------------------------------------  
$ERROR 
 
IPRED = 0 
IF (TICLO.EQ.0.AND.CMT.EQ.2) THEN 
IPRED = F 
Y = IPRED*(1+EPS(1)) 
ENDIF 
 
IF (TICLO.EQ.0.AND.CMT.EQ.5) THEN 
IPRED = F 
Y = IPRED*(1+EPS(2)) 



ENDIF 
 
IF (TICLO.EQ.1.AND.CMT.EQ.2) THEN 
IPRED = F 
Y = IPRED*(1+EPS(3)) 
ENDIF 
 
IF (TICLO.EQ.1.AND.CMT.EQ.5) THEN 
IPRED = F 
Y = IPRED*(1+EPS(4)) 
ENDIF 
 
IF (TICLO.EQ.1.AND.CMT.EQ.11) THEN 
IPRED = F 
Y = IPRED*(1+EPS(5)) 
ENDIF 
 
IRES = DV - IPRED 
DEL = 0 
IF(IPRED.EQ.0) DEL = 1 
IWRES = (1-DEL)*IRES/(IPRED+DEL) 
 
C2 = A(2)/KV1 
C3 = A(3)/KV2 
C4 = A(4)/KV3 
C5 = A(5)/NKV5 
C6 = A(6)/NKV6 
C8 = A(8)/VGW 
C9 = A(9)/VPV 
C10 = A(10)/VH 
C11 = A(11)/TV1 
C12 = A(12)/TV2 
 
;----------------------------------------------------------------------------------  
$THETA 
 
(0,13.5)                     ; KV1 
(0,265.2)                    ; KQ 
(0,104.8)                    ; KV2 
(0,1.91)                     ; KKA 
(0,303.6)                    ; KCLINTH1 
(0,174.5)                    ; KV3 
(0,21.8)                     ; KQ2 
(0,1.16)                     ; KCLINTG1 
4.1 FIX                      ; KCLP 
(0,87.9)                     ; NKV5 
(0,20.9)                     ; NKQ3 
(0,85.5)                     ; NKV6 
(0,78.1)                     ; NKCLINTH 
(0,45.1)                     ; NKCLINTG 
3.3 FIX                      ; TKA 
(0,1500)                     ; TCLINTH 
(0,50)                       ; TV1 
(0,25)                       ; TQ 
(0,200)                      ; TV2 
0 FIX                        ; TCLINTG 
0.57 FIX                     ; TKI 
18 FIX                       ; TKINACT 
0.63 FIX                     ; KFM2B6 
0.017 FIX                    ; LAMBDA 
0.031 FIX                    ; TKIN 
 
;----------------------------------------------------------------------------------  
$OMEGA 



 
0.05                         ; KV2_ 
0.42                         ; KKA_ 
0.25                         ; KCLINTH1_ 
1.99                         ; KCLINTG1_ 
0.10                         ; NKCLINTH_ 
0.13                         ; TCLINTH_ 
 
;----------------------------------------------------------------------------------  
$SIGMA 
 
0.086 
0.062 
0.065 
0.066 
0.066 
 
;----------------------------------------------------------------------------------  
$ESTIMATION METHOD=COND INTERACTION NSIG=2 SIGL=6 PRINT=5 MAXEVAL=9999 NOABORT 
POSTHOC MSFO=msfo.outputfile 
; $SIMULATION (123456) ONLYSIM PREDICTION SUBPROBS=1000 TRUE=INITIAL 
 
;----------------------------------------------------------------------------------  
$COVARIANCE MATRIX=S PRINT=E 
 
;----------------------------------------------------------------------------------  
$TABLE 
ID TIME AMT MDV DV IPRED CMT EVID NPD NPDE IWRES CWRES 
NOPRINT ONEHEADER 
FILE=sdtab1 
 
;----------------------------------------------------------------------------------  
$TABLE 
ID TIME DV PRED IPRED WTKG KKA KV1 KQ KV2 KCLINTH1 CLINTH 
CLH KCLINTG1 CLINTG CLG KV3 KQ2 NKV5 NKQ3 NKV6 NKCLINTH 
CLHM NKCLINTG CLINTGM CLGM TKA TCLINTH TV1 TQ TV2 TCLINTG  
KCLINTH1 CLINTH TCE LAMBDA TAH 
TBH ZETA INHBH 
ETA1 ETA2 ETA3 ETA4 ETA5 ETA6 
NOPRINT ONEHEADER FILE=patab1 
 
;----------------------------------------------------------------------------------  
$TABLE 
ID TIME WTKG C2 C3 C4 C5 C6 C8 C9 C10 C11 C12 NOPRINT 
ONEHEADER 
FILE=mytab1 


